
Flop is an Elixir library designed to easily apply filtering, ordering, and
pagination to your Ecto queries.

  
    
  
  Features


	Offset-based pagination: Allows pagination through offset/limit or
page/page_size parameters.
	Cursor-based pagination: Also known as key set pagination, provides a more
efficient alternative to offset-based pagination. Compatible with Relay
pagination arguments.
	Sorting: Applies sort parameters on multiple fields in any direction.
	Filtering: Allows complex data filtering using multiple conditions,
operators, and fields.
	Parameter validation: Ensures the validity of provided parameters.
	Configurable filterable and sortable fields: Only applies parameters to
the fields that were explicitly configured as filterable or sortable.
	Join fields: Allows the application of pagination, sort, and filter
parameters on any named binding. Provides functions to help you to avoid
unnecessary join clauses.
	Compound fields: Provides the ability to apply filter parameters on
multiple string fields, for example for a full name filter.
	Custom fields: Provides an escape hatch for filters that Flop is not able
to build on its own.
	Relay connection formatter: Formats the connection in Relay style,
providing edges, nodes, and page info.
	UI helpers and URL builders through
Flop Phoenix: Pagination, sortable
tables and filter forms.


  
    
  
  Installation


To get started, add flop to your dependencies list in your project's mix.exs
file:
def deps do
  [
    {:flop, "~> 0.26.0"}
  ]
end
You can also configure a default repo for Flop by adding the following line to
your config file:
config :flop, repo: MyApp.Repo
Instead of configuring Flop globally, you can also use a configuration module.
Please refer to the Flop module documentation for more information.

  
    
  
  Usage



  
    
  
  Define sortable and filterable fields


To define sortable and filterable fields in your Ecto schema, you can derive
Flop.Schema. This step is optional but highly recommended, particularly when
the parameters passed to Flop's functions are user-provided. Deriving
Flop.Schema ensures that Flop applies filtering and sorting parameters only to
the fields you've explicitly configured.
defmodule MyApp.Pet do
  use Ecto.Schema

  @derive {
    Flop.Schema,
    filterable: [:name, :species],
    sortable: [:name, :age, :species]
  }

  schema "pets" do
    field :name, :string
    field :age, :integer
    field :species, :string
    field :social_security_number, :string
  end
end
Besides sortable and filterable fields, Flop.Schema also allows the definition
of join fields, compound fields, or custom fields. You can also set maximum or
default limits, among other options. For a comprehensive list of available
options, check the Flop.Schema documentation.

  
    
  
  Query data


Use the Flop.validate_and_run/3 or Flop.validate_and_run!/3 function to both
validate the parameters and fetch data from the database, and acquire pagination
metadata in one operation.
Here is an example of how you might use this in your code:
defmodule MyApp.Pets do
  import Ecto.Query, warn: false

  alias Ecto.Changeset
  alias MyApp.{Pet, Repo}

  @spec list_pets(map) ::
          {:ok, {[Pet.t()], Flop.Meta.t()}} | {:error, Flop.Meta.t()}
  def list_pets(params \\ %{}) do
    Flop.validate_and_run(Pet, params, for: Pet)
  end
end
The for option sets the Ecto schema for which you derived Flop.Schema. If
you haven't derived Flop.Schema as described above, this option can be
omitted. However, this is not recommended unless all parameters are generated
internally and are guaranteed to be safe.
On success, Flop.validate_and_run/3 returns an :ok tuple. The second element
of this tuple is another tuple containing the fetched data and metadata.
{:ok, {[%Pet{}], %Flop.Meta{}}}
You can learn more about the Flop.Meta struct in the
module documentation.
Alternatively, you may separate parameter validation and data fetching into
different steps using the Flop.validate/2, Flop.validate!/2, and Flop.run/3
functions. This allows you to manipulate the validated parameters, to modify the
query depending on the parameters, or to move the parameter validation to a
different layer of your application.
with {:ok, flop} <- Flop.validate(params, for: Pet) do
  Flop.run(Pet, flop, for: Pet)
end
The aforementioned functions internally call the lower-level functions
Flop.all/3, Flop.meta/3, and Flop.count/3. If you have advanced
requirements, you might prefer to use these functions directly. However, it's
important to note that these lower-level functions do not validate the
parameters. If parameters are generated based on user input, they should always
be validated first using Flop.validate/2 or Flop.validate!/2 to ensure safe
execution.
The examples above assume that you configured a default repo. However, you can
also pass the repo directly to the functions:
Flop.validate_and_run(Pet, flop, repo: MyApp.Repo)
Flop.all(Pet, flop, repo: MyApp.Repo)
Flop.meta(Pet, flop, repo: MyApp.Repo)
For more detailed information, refer the
documentation.

  
    
  
  Parameter format


The Flop library requires parameters to be provided in a specific format as a
map. This map can be translated into a URL query parameter string, typically
for use in a web framework like Phoenix.

  
    
  
  Pagination


Offset / limit
You can specify an offset to start from and a limit to the number of results.
%{offset: 20, limit: 10}
This translates to the following query parameter string:
?offset=20&limit=10
Page / page size
You can specify the page number and the size of each page.
%{page: 2, page_size: 10}
This translates to the following query parameter string:
?page=2&page_size=10
Cursor
You can fetch a specific number of results before or after a given cursor.
%{first: 10, after: "g3QAAAABZAACaWRiAAACDg=="}
%{last: 10, before: "g3QAAAABZAACaWRiAAACDg=="}
These translate to the following query parameter strings:
?first=10&after=g3QAAAABZAACaWRiAAACDg==
?last=10&before=g3QAAAABZAACaWRiAAACDg==

  
    
  
  Ordering


To sort the results, specify fields to order by and the direction of sorting for
each field.
%{order_by: [:name, :age], order_directions: [:asc, :desc]}
This translates to the following query parameter string:
?order_by[]=name&order_by[]=age&order_directions[]=asc&order_directions[]=desc

  
    
  
  Filters


You can filter the results by providing a field, an operator, and a value. The
operator is optional and defaults to ==. Multiple filters are combined with a
logical AND. At the moment, combining filters with OR is not supported.
%{filters: [%{field: :name, op: :ilike_and, value: "Jane"}]}
This translates to the following query parameter string:
?filters[0][field]=name&filters[0][op]=ilike_and&filters[0][value]=Jane
Refer to the Flop.Filter documentation and Flop.t/0 type documentation for
more details on using filters.

  
    
  
  Internal parameters


Flop is designed to manage parameters that come from the user side. While it is
possible to alter those parameters and append extra filters upon receiving them,
it is advisable to clearly differentiate parameters coming from outside and the
parameters that your application adds internally.
Consider the scenario where you need to scope a query based on the current user.
In this case, it is better to create a separate function that introduces the
necessary WHERE clauses:
def list_pets(%{} = params, %User{} = current_user) do
  Pet
  |> scope(current_user)
  |> Flop.validate_and_run(params, for: Pet)
end

defp scope(q, %User{role: :admin}), do: q
defp scope(q, %User{id: user_id}), do: where(q, user_id: ^user_id)
If you need to add extra filters that are only used internally and aren't
exposed to the user, you can pass them as a separate argument. This same
argument can be used to override certain options depending on the context in
which the function is called.
def list_pets(%{} = params, opts \\ [], %User{} = current_user) do
  flop_opts =
    opts
    |> Keyword.take([
      :default_limit,
      :default_pagination_type,
      :pagination_types
    ])
    |> Keyword.put(:for, Pet)

  Pet
  |> scope(current_user)
  |> apply_filters(opts)
  |> Flop.validate_and_run(params, flop_opts)
end

defp scope(q, %User{role: :admin}), do: q
defp scope(q, %User{id: user_id}), do: where(q, user_id: ^user_id)

defp apply_filters(q, opts) do
  Enum.reduce(opts, q, fn
    {:last_health_check, dt}, q -> where(q, [p], p.last_health_check < ^dt)
    {:reminder_service, bool}, q -> where(q, [p], p.reminder_service == ^bool)
    _, q -> q
  end)
end
With this approach, you maintain a clean separation between user-driven
parameters and system-driven parameters, leading to more maintainable and less
error-prone code.

  
    
  
  Relay and Absinthe


The Flop.Relay module is useful if you are using
absinthe with
absinthe_relay, or if you simply need
to adhere to the Relay cursor specification. This module provides functions that
help transform query responses into a format compatible with Relay.
Consider the scenario where you have defined node objects for owners and pets,
along with a connection field for pets on the owner node object.
node object(:owner) do
  field :name, non_null(:string)
  field :email, non_null(:string)

  connection field :pets, node_type: :pet do
    resolve &MyAppWeb.Resolvers.Pet.list_pets/2
  end
end

node object(:pet) do
  field :name, non_null(:string)
  field :age, non_null(:integer)
  field :species, non_null(:string)
end

connection(node_type: :pet)
Absinthe Relay will establish the arguments after, before, first and
last on the pets field. These argument names align with those used by Flop, facilitating their application.
Next, we'll define a list_pets_by_owner/2 function in the Pets context.
defmodule MyApp.Pets do
  import Ecto.Query

  alias MyApp.{Owner, Pet, Repo}

  @spec list_pets_by_owner(Owner.t(), map) ::
          {:ok, {[Pet.t()], Flop.Meta.t()}} | {:error, Flop.Meta.t()}
  def list_pets_by_owner(%Owner{id: owner_id}, params \\ %{}) do
    Pet
    |> where(owner_id: ^owner_id)
    |> Flop.validate_and_run(params, for: Pet)
  end
end
Now, within your resolver, you merely need to invoke the function and call
Flop.Relay.connection_from_result/1, which transforms the result into a tuple
composed of the edges and the page_info, as required by absinthe_relay.
defmodule MyAppWeb.Resolvers.Pet do
  alias MyApp.{Owner, Pet}

  def list_pets(args, %{source: %Owner{} = owner} = resolution) do
    with {:ok, result} <- Pets.list_pets_by_owner(owner, args) do
      {:ok, Flop.Relay.connection_from_result(result)}
    end
  end
end
In case you want to introduce additional filter arguments, you can employ
Flop.nest_filters/3 to convert simple filter arguments into Flop filters,
without necessitating API users to understand the Flop filter format.
Let's add name and species filter arguments to the pets connection field.
node object(:owner) do
  field :name, non_null(:string)
  field :email, non_null(:string)

  connection field :pets, node_type: :pet do
    arg :name, :string
    arg :species, :string

    resolve &MyAppWeb.Resolvers.Pet.list_pets/2
  end
end
Assuming that these fields have been already configured as filterable with
Flop.Schema, we can use Flop.nest_filters/3 to take the filter arguments and
transform them into a list of Flop filters.
defmodule MyAppWeb.Resolvers.Pet do
  alias MyApp.{Owner, Pet}

  def list_pets(args, %{source: %Owner{} = owner} = resolution) do
    args = nest_filters(args, [:name, :species])

    with {:ok, result} <- Pets.list_pets_by_owner(owner, args) do
      {:ok, Flop.Relay.connection_from_result(result)}
    end
  end
end
Flop.nest_filters/3 uses the equality operator :== by default.
You can override the default operator per field.
args = nest_filters(args, [:name, :species], operators: %{name: :ilike_and})

  
    
  
  Flop Phoenix


Flop Phoenix is a companion library that
provides Phoenix components for pagination, sortable tables, and filter forms,
usable with both Phoenix LiveView and in dead views. It also defines helper
functions to build URLs with Flop query parameters.

  Unreleased



  
    
  
  [0.26.0] - 2024-08-18



  
    
  
  Removed


	The previously deprecated tuple syntax for defining join fields has been
removed in favor of a keyword list.
	The previously deprecated function Flop.Schema.field_type/2 was removed in
favor of Flop.Schema.field_info/2.


  
    
  
  Fixed


	Fixed a compatibility issue with Ecto 3.12 related to the initialization of
the Ecto.Enum type.


  
    
  
  Upgrade Guide


Replace the tuple syntax for join fields with a keyword list.
@derive {
  Flop.Schema,
  join_fields: [
-    owner_name: {:owner, :name}
+    owner_name: [
+      binding: :owner,
+      field: :name
+    ]
  ]
}

  
    
  
  [0.25.0] - 2024-01-14



  
    
  
  Added


	Added Flop.Filter.update_value/3 for updating the filter value for a field
in a list of filters.


  
    
  
  Fixed


	Determine pagination type if pagination parameter has errors.


  
    
  
  [0.24.1] - 2023-11-18



  
    
  
  Changed


	Flop.push_order/3 now allows you to use a descending order as the initial
sort order.


  
    
  
  [0.24.0] - 2023-11-14



  
    
  
  Changed


	If an invalid operator is passed in a filter, the error will now include the
list of allowed operators for that field.


  
    
  
  [0.23.0] - 2023-09-26



  
    
  
  Added


	Added directions option to Flop.push_order/3.


  
    
  
  Fixed


	Escape backlash character in queries using one of the like operators.


  
    
  
  [0.22.1] - 2023-07-18



  
    
  
  Fixed


	Updated version requirement for Ecto to ~> 3.10.3. Flop 0.22.0 relies
on a feature added in that version and doesn't compile with lower versions.


  
    
  
  [0.22.0] - 2023-07-17


This release includes a substantial refactoring to lay the groundwork for the
upcoming adapter feature. While this release contains deprecations and changes,
they are either backward compatible or affect functions that are unlikely to be
used by end users. The primary aim has been to ensure a seamless transition and
maintain compatibility with previous versions.

  
    
  
  Added


	Added a Flop.FieldInfo struct that contains metadata for a field for use
by adapters.
	Added the Flop.Schema.field_info/2 function, which derives field information
and replaces the previous Flop.Schema.field_type/2 function with a more
standardized and structured output.


  
    
  
  Changed


	The Ecto-specific options alias_fields, compound_fields, custom_fields,
and join_fields within Flop.Schema, as well as repo and query_opts
within use Flop, are now nested under the adapter_opts keyword. The old
configuration format is still supported.


  
    
  
  Deprecated


	Flop.Schema.field_type/2 was deprecated in favor of
Flop.Schema.field_info/2.


  
    
  
  Removed


	Removed Flop.Schema.apply_order_by/3.
	Removed Flop.Schema.cursor_dynamic/3.


  
    
  
  Upgrade guide


While the old configuration format is still supported, you are invited to
update your application to the new structure to prepare for future versions.
To do this, place the field configuration for Flop.Schema under
adapter_opts:
@derive {
  Flop.Schema,
  filterable: [],
  sortable: [],
-  alias_fields: [],
-  compound_fields: [],
-  custom_fields: [],
-  join_fields: []
+  adapter_opts: [
+    alias_fields: [],
+    compound_fields: [],
+    custom_fields: [],
+    join_fields: []
+  ]
}
Similarly for use Flop, you can nest repo and query_opts under
adapter_opts:
use Flop,
  default_limit: 50,
-  repo: MyApp.Repo,
-  query_opts: [prefix: "some-prefix"]
+  adapter_opts: [
+    repo: MyApp.Repo,
+    query_opts: [prefix: "some-prefix"]
+  ]

  
    
  
  [0.21.0] - 2023-07-02



  
    
  
  Added


	Introduced operators as a new option for restricting acceptable operators
for a custom field.
	Added bindings option for custom fields, allowing required named bindings to
be added via Flop.with_named_bindings/4.
	The ecto_type option on join and custom fields now supports
references: {:from_schema, MySchema, :some_field}.
	The ecto_type option now supports a convenient syntax for adhoc enums:
{:ecto_enum, [:one, :two]}.
	Improved documentation with added type definitions: Flop.Schema.option/0,
Flop.Schema.join_field_option/0, Flop.Schema.custom_field_option/0,
and Flop.Schema.ecto_type/0, describing options available when deriving
the Flop.Schema protocol.


  
    
  
  Changed


	Breaking change: Filter values are now dynamically cast based on the
field type and operator, instead of allowing any arbitrary filter value. This
change ensures that invalid filter values cause validation errors instead of
cast errors.
	The options for deriving the Flop.Schema protocol and for use Flop
now undergo stricter validation with NimbleOptions.
	Flop.Cursor.encode/1 now explicitly sets the minor version option for
:erlang.term_to_binary/2 to 2, aligning with the new default in OTP 26.
Before, this option was not set at all.
	Added a decoded_cursor field to the Flop struct. This field temporarily
stores the decoded cursor between validation and querying and is
discarded when generating the meta data.


  
    
  
  Deprecated


	The tuple syntax for defining join fields has been deprecated in favor of a
keyword list.


  
    
  
  Fixed


	Resolved an issue where setting replace_invalid_params to true still
caused validation errors for pagination and sorting parameters due to cast
errors, instead of defaulting to valid parameters.
	Fixed the type specification for Flop.Filter.allowed_operators/1.


  
    
  
  Upgrade notes


The newly implemented dynamic casting of filter values could impact your code:
	Filter values failing to cast into the determined type will now yield a
validation error or result in the removal of the invalid filter if the
replace_invalid_params option is enabled.
	The value field of the Flop.Filter struct now holds the cast value
instead of the original parameter value. For instance, while handling
parameters generated via an HTML form with Flop, previously all filter values
would be represented as strings in the struct. However, they may now be
integers, DateTime structs, and so forth. Look out for this if you are
directly reading or manipulating Flop.Filter structs.
	For join and custom fields, the type is determined with the ecto_type
option. Previously, this option was only used for operator validation.
Ensure the correct Ecto type is set. If the option is omitted, the filter
values will continue to use their incoming format.
	Manual casting of filter values in a custom filter function is no longer
required if the ecto_type option is set.
	If join fields point to Ecto.Enum fields, previously you could simply set
ecto_type to string. This will continue to work if the filter value is
passed as a string, but passing it as an atom will cause an error. Make sure
to correctly reference the schema field
({:from_schema, MySchema, :some_field}) or directly pass the Enum values
({:ecto_enum, [:one, :two}).
	To enable Flop.Phoenix to build a query string for filter parameters, the
filter value must be convertible into a string via to_string/1. If
ecto_type is set to a custom Ecto type that casts values into a struct, the
String.Chars protocol must be implemented for that struct.
	If you use the result of Flop.Phoenix.to_query/2 in a ~p sigil for
verified routes or in a route helper function, Phoenix converts filter values
into a string using the Phoenix.Param protocol. If you use Date,
DateTime, NaiveDateTime, Time filters, or filters using custom structs,
you need to implement that protocol for these structs in your application.

Please review the newly added "Ecto type option" section in the Flop.Schema
module documentation.
Join field syntax
If you are using tuples to define join fields when deriving Flop.Schema,
update the configuration to use keyword lists instead:
@derive {
  Flop.Schema,
  join_fields: [
-    owner_name: {:owner, :name}
+    owner_name: [binding: :owner, field: :name]
  ]
}

  
    
  
  [0.20.3] - 2023-06-23



  
    
  
  Changed


	Flop.count/3 will now wrap queries that have GROUP BY clauses in a
subquery.


  
    
  
  Fixed


	Fixed cursor-based pagination on composite types.


  
    
  
  [0.20.2] - 2023-06-09



  
    
  
  Changed


	Added nutrition facts about use Flop and @derive Flop.Schema.
	The minimum Elixir version is now 1.11.


  
    
  
  Fixed


	Fixed a deprecation warning about Logger.warn/1.
	Fixed a deprecation warning about passing an MFA to :with in
cast_assoc/cast_embed introduced in Ecto 3.10.2.


  
    
  
  [0.20.1] - 2023-05-19



  
    
  
  Added


	Added the :count override option to Flop.count/3.


  
    
  
  Changed


	The default_pagination_type can now be set in the schema.


  
    
  
  Fixed


	Don't raise function clause error in Flop.to_previous_cursor/1 and
Flop.to_next_cursor/1 when the start cursor or end cursor are nil.


  
    
  
  [0.20.0] - 2023-03-21



  
    
  
  Added


	Added Flop.unnest_filters/3 as a reverse operation of Flop.nest_filters/3
after retrieving data from the database.
	Added Flop.Filter.fetch_value/2, Flop.Filter.get_value/2,
Flop.Filter.put_value/4, Flop.Filter.put_new_value/4,
Flop.Filter.pop_value/3 and Flop.Filter.pop_first_value/3.


  
    
  
  Changed


	Several of the functions for manipulating lists of filters in the
Flop.Filter module now accept lists of maps with atom keys, lists of maps
with string keys, and indexed maps as produced by Phoenix HTML forms as
argument.
	The empty and not_empty operators now treat empty maps as empty values on
map fields and empty arrays as empty values on array fields.
	% and _ characters in filter values for the like, ilike and =~
operators are now escaped.


  
    
  
  Fixed


	Fixed an issue that caused filter conditions for like_and, like_or,
ilike_and and ilike_or to be incorrectly combined when applied to compound
fields.


  
    
  
  [0.19.0] - 2023-01-15



  
    
  
  Added


	Support for custom fields. These fields allow you to run custom filter
functions for anything that cannot be expressed with Flop filters.
	Added Flop.with_named_bindings/4 for dynamically adding bindings needed for
a Flop query.
	Added fetch, get, get_all, delete, delete_first, drop, new,
take, pop, pop_first, put and put_new functions to Flop.Filter.
	Added Flop.Meta.with_errors/3.
	Added ecto_type option to join fields.
	Added not_like and not_ilike filter operators.
	Added a cheatsheet for schema configuration.
	Added opts field to Flop.Meta struct.


  
    
  
  Changed


	Renamed Flop.bindings/3 to Flop.named_bindings/3.
	Flop.Filter.allowed_operators/2 now tries to determine the Ecto type by
reading the Flop field type from the schema module. This function is used
during parameter validation, which means the validation step will be a bit
stricter now. For join and custom fields, the Ecto type is determined via the
new ecto_type option. If the option is not set, the function returns all
operators as before. For compound fields, only the supported operators are
returned.


  
    
  
  [0.18.4] - 2022-11-17



  
    
  
  Changed


	The :ilike_and, :ilike_or, :like_and and :like_or filter operators can
now also be used with a list of strings as filter value.


  
    
  
  [0.18.3] - 2022-10-27



  
    
  
  Fixed


	default_pagination_type can be overridden by passing false now.


  
    
  
  [0.18.2] - 2022-10-19



  
    
  
  Fixed


	Flop.bindings/3 did not consider join fields that are used as part of a
compound field.


  
    
  
  [0.18.1] - 2022-10-14



  
    
  
  Changed


	If the given map already has a :filters / "filters" key,
Flop.nest_filters/3 will now merge the derived filters into the existing
filters. If the existing filters are formatted as a map (as produced by an
HTML form), they are converted to a list first.
	use Flop will now also compile validate/2 and validate!/2 functions that
apply the options of your config module.
	Allow setting default_limit and max_limit to false, which removes the
default/max limit without falling back to global options.


  
    
  
  Fixed


	Flop.bindings/3 was returning bindings for filters with nil values.


  
    
  
  [0.18.0] - 2022-10-10



  
    
  
  Added


	Added alias_fields option to Flop.Schema, which allows you to sort by
field aliases defined with Ecto.Query.API.selected_as/2.
	Added aliases/2 for getting the alias fields needed for a query.
	Added documentation example for filtering by calculated values.
	New option rename for Flop.map_to_filter_params/2 and
Flop.nest_filters/3.
	New option :replace_invalid_params. This option can be passed to the
validate and validate_and_run functions or set in the global configuration
or in a config module. Setting the value to true will cause Flop to replace
invalid parameters with default values where possible or remove the parameter
otherwise during the validation step, instead of returning validation errors.


  
    
  
  Changed


	Require ecto ~> 3.9.0.
	Flop.Schema does not raise an error anymore if a compound or join field is
defined with the same name as a regular Ecto schema field. This was done so that
you can add virtual fields with the same name. It is not possible to
differentiate between non-virtual and virtual fields at compile time (at least
I don't know how), so we cannot differentiate in the validation step.
	Flop applies a default limit of 50 and a max limit of 1000 now, unless
other values are set.
	In offset/limit based pagination, the limit parameter is now required, in
line with the other pagination types. If not set, it will fall back to a
default limit.


  
    
  
  [0.17.2] - 2022-10-03



  
    
  
  Fixed


	Fixed an issue where the repo option was not read from a backend module.


  
    
  
  [0.17.1] - 2022-10-02



  
    
  
  Added


	Added a backend field to the Flop.Meta struct.


  
    
  
  Fixed


	Fixed an issue where the schema options were overridden by the backend module
options.


  
    
  
  [0.17.0] - 2022-08-26



  
    
  
  Added


	Added the filter operators not_in and not_contains.
	Added examples for integration with Relay to the documentation.
	Added examples for the parameter format to the documentation.


  
    
  
  Changed


	Refactored the query builder. This does not affect users of the library, but
makes the code base more readable and lays the groundwork for upcoming
features.
	Added the :query_opts option to Flop callbacks to pass on options to the
Ecto repo on query execution. If you are already using the :prefix option
you now have to pass this through :query_opts.

If you configured the Repo :prefix in the application config:
config :flop,
-  prefix: "some-prefix"
+  query_opts: [prefix: "some-prefix"]
If you set the :prefix when calling the Flop functions:
- Flop.validate_and_run(Pet, params, prefix: "some-prefix")
+ Flop.validate_and_run(Pet, params, query_opts: [prefix: "some-prefix"])

  
    
  
  [0.16.1] - 2022-04-05



  
    
  
  Fixed


	Wrong type spec for Flop.Schema.default_order/1 callback.


  
    
  
  [0.16.0] - 2022-03-22



  
    
  
  Added


	You can now define a configuration module with use Flop to set defaults
instead of or in addition to the application configuration. This makes it
easier to work with multiple Ecto repos.
	The new function Flop.bindings/3 returns the necessary bindings for a
given Flop query. You can use it in case you want to optimize your queries by
only joining tables that are actually needed.
	Added a count_query option to override the count query used by
Flop.run/3, Flop.validate_and_run/3 and Flop.validate_and_run!/3.
	You can get a list of allowed operators for a given Ecto type or a given
schema field with Flop.Filter.allowed_operators/1 and
Flop.Filter.allowed_operators/2 now.


  
    
  
  Changed


	Breaking: The :empty and :not_empty filters now require a boolean value.
If no value is passed, the filter is ignored, just as it is handled for all
other filter operators. This change was necessary to make the integration
with filter forms (checkboxes) easier.
	Breaking: The default order needs to be passed as a map now when deriving
Flop.Schema. The previous implementation already converted the two separate
configuration keys to a map. This meant that the configuration passed when
deriving Flop.Schema had a different format from the one you had to pass
when overriding the default order with the opts.
With this change, the configuration format is the same everywhere. A compile
time exception is raised if you are still using the old format, guiding you in
the update.
	It is now validated that the filter operator matches the field type.
	The compile time validation of the options passed when deriving Flop.Schema
has been improved.
	Allow passing page as string to Flop.set_page/2.
	Allow passing offset as string to Flop.set_offset/2.


  
    
  
  [0.15.0] - 2021-11-14



  
    
  
  Added


	Add Flop.reset_filters/1 and Flop.reset_order/1.
	Add Flop.current_order/2 to retrieve the order of a given field.
	Add Flop.to_next_page/2 and Flop.to_previous_page/1.
	Add Flop.set_cursor/2, Flop.to_next_cursor/1 and
Flop.to_previous_cursor/1.
	Add Flop.set_offset/2, Flop.to_previous_offset/1, Flop.to_next_offset_2
and Flop.reset_cursors/2.
	Add Flop.nest_filters/3 for converting filters between a key/value map and
a list of Flop.Filter parameters.
	You can now set the default_pagination_type option, which forces a certain
set of parameters when defaults are applied and the pagination type cannot
be determined from the given parameters.
	Add optional default argument to get_option.
	Add pagination option. If set to true, pagination parameters are not cast.


  
    
  
  Changed


	Flop.map_to_filter_params/2 returns maps with string keys if the original
map has string keys now.
	The has_previous_page? value of the Flop.Meta struct is now always true
if first is used with after. has_next_page? is always true when
last is used with before.
	push_order/2 resets the :after and :before parameters now, since the
cursors depend on the order.
	validate_and_run/3 and validate_and_run!/3 pass all given options to
the validate functions now, allowing you to override defaults set in the
schema.
	If the pagination_types option is used, parameters for other pagination
types will not be cast now instead of casting them and returning validation
errors.


  
    
  
  Removed


	Remove Flop.Cursor.get_cursor_from_map/2. Use
Flop.Cursor.get_cursor_from_node/2 instead.


  
    
  
  [0.14.0] - 2021-11-08



  
    
  
  Added


	Add :contains operator.
	Add Flop.map_to_filter_params/2.


  
    
  
  Changed


	Flop.validate/2 and Flop.validate_and_run/3 return {:error, Flop.Meta.t}
instead of {:error, Ecto.Changeset.t} now. The Meta struct has the new
fields :errors and :params, which are set when validation errors occur.
This accompanies the changes in Flop.Phoenix, which include the
implementation of the Phoenix.HTML.FormData protocol for the Flop.Meta
struct.
	Flop.validate!/2 and Flop.validate_and_run!/3 raise a
Flop.InvalidParamsError instead of an Ecto.InvalidChangesetError now.
	Add :schema key to Flop.Meta. This field points to the schema module set
by passing the :for option.
	Minimum Ecto version changed to 3.5.
	Replace Operator and OrderDirection custom Ecto types with Ecto.Enum.
	Update Flop.Meta struct default values for the fields :flop,
:has_next_page? and :has_previous_page?.


  
    
  
  [0.13.2] - 2021-10-16



  
    
  
  Fixed


	Fix error when sorting by a compound field that consists of at least one
join field.
	Fix import conflict when importing Ecto.Changeset in a module that derives
Flop.Schema and configures a compound field.


  
    
  
  [0.13.1] - 2021-08-23



  
    
  
  Fixed


	Wrong type spec for cursor_dynamic/3 callback.


  
    
  
  [0.13.0] - 2021-08-22



  
    
  
  Added


	Support ordering by join fields.
	Support ordering by compound fields.
	Support join fields as cursor fields.
	New function Flop.Schema.get_field/2.
	Flop.Cursor.get_cursor_from_edge/2 and Flop.Cursor.get_cursor_from_node/2
can get cursor values from join and compound fields now.


  
    
  
  Changed


To get the pagination cursor value from a join field, Flop needs to know how
to access the field value from the returned struct or map. The configuration
format for join fields has been changed to allow specifying the path to the
nested field.
Before:
@derive {
  Flop.Schema,
  join_fields: [
    owner_name: {:owner, :name}
  ]
}
After:
@derive {
  Flop.Schema,
  join_fields: [
    owner_name: [binding: :owner, field: :name, path: [:owner, :name]]
  ]
}
The :path is optional and inferred from the :binding and :field options,
if omitted.
The old configuration format is still accepted. All of these settings are
equivalent:
[owner_name: {:owner, :name}]

[owner_name: [binding: :owner, field: :name]]

[owner_name: [binding: :owner, field: :name, path: [:owner, :name]]]

  
    
  
  Fixed


	Cursor pagination failed when one of the cursor field values was nil.


  
    
  
  [0.12.0] - 2021-08-11



  
    
  
  Added


	Allow to define join fields in Flop.Schema.
	Allow to define compound fields in Flop.Schema.
	Support filtering by join fields.
	Support filtering by compound fields.
	New filter operator empty.
	New filter operator not_empty.
	New function Flop.set_page/2.


  
    
  
  Changed


	Rename option get_cursor_value_func to cursor_value_func.
	Silently ignore filters with nil value for the field or the value instead of
raising an ArgumentError.
	Allow passing a string as the second argument to Flop.push_order/2.


  
    
  
  [0.11.0] - 2021-06-13



  
    
  
  Added


	New functions Flop.Cursor.get_cursor_from_node/2 and
Flop.Cursor.get_cursor_from_edge/2.
	New function Flop.get_option/2.
	Support Ecto prefixes.


  
    
  
  Changed


	Use Flop.Cursor.get_cursor_from_node/2 as default for the
:get_cursor_value_func option.
	Flop.Relay.edges_from_result/2 can now handle nil instead of a map as
edge information in a query result.


  
    
  
  Deprecated


	Deprecate Flop.Cursor.get_cursor_from_map/2. Use
Flop.Cursor.get_cursor_from_node/2 instead.


  
    
  
  [0.10.0] - 2021-05-03



  
    
  
  Added


	Add function Flop.push_order/2 for updating the order_by and
order_directions values of a Flop struct.


  
    
  
  [0.9.1] - 2020-10-21



  
    
  
  Fixed


	Fixed type spec of Flop.Schema.default_order/1.


  
    
  
  [0.9.0] - 2020-10-16



  
    
  
  Added


	Add like, like_and, like_or, ilike, ilike_and and ilike_or filter
operators.
	Add option to disable pagination types globally, for a schema or locally.
	Add options to disable ordering or filtering.
	Allow global configuration of get_cursor_value_func, max_limit and
default_limit.
	Add Flop.option type, improve documentation of available options.
	Add Flop.Cursor.decode!/1.


  
    
  
  Changed


	Refactored the parameter validation. Default limits are now applied to all
pagination types. Added validation for the after / before cursor values.
	Flop.Cursor.decode/1 returns :ok tuple or :error now instead of raising
an error if the cursor is invalid.
	Flop.Cursor.decode/1 returns an error if the decoded cursor value is not a
map with atom keys.
	Improved documentation.


  
    
  
  [0.8.4] - 2020-10-14



  
    
  
  Fixed


	Default limit was overriding first / last parameters when building query.


  
    
  
  [0.8.3] - 2020-10-14



  
    
  
  Fixed


	Cursor-based pagination: has_next_page? was set when querying with last
based on before being set. Likewise, has_previous_page? was set when
querying with first based on after being set. Both assumptions are wrong.
In both cases, the values are always set to false now.


  
    
  
  [0.8.2] - 2020-10-08



  
    
  
  Changed


	Order directions are not restricted anymore for cursor-based pagination.


  
    
  
  Fixed


	Query for cursor-based pagination returned wrong results when using more than
one cursor field.
	Query for cursor-based pagination returned wrong results when using
last/before.


  
    
  
  [0.8.1] - 2020-10-07



  
    
  
  Changed


	Allow structs in cursor values.


  
    
  
  [0.8.0] - 2020-10-07



  
    
  
  Added


	Support for cursor-based pagination. Thanks to @bunker-inspector.
	Add functions to turn query results into Relay connection format when using
cursor-based pagination.


  
    
  
  [0.7.1] - 2020-09-04



  
    
  
  Fixed


	Calculation of has_next_page? was wrong.


  
    
  
  [0.7.0] - 2020-08-04



  
    
  
  Added


	Flop.Schema now allows to set a default sort order.


  
    
  
  Changed


	Passing a limit without an offset will now set the offset to 0.
	Passing a page size without a page will now set the page to 1.


  
    
  
  [0.6.1] - 2020-06-17



  
    
  
  Changed


	Add Flop to Meta struct.


  
    
  
  Fixed


	Type Flop.Filter.op didn't include all operators.


  
    
  
  [0.6.0] - 2020-06-14



  
    
  
  Added


	New struct Flop.Meta.
	New function Flop.all/3.
	New function Flop.count/3.
	New function Flop.meta/3.
	New function Flop.run/3.
	New function Flop.validate_and_run/3.
	New function Flop.validate_and_run!/3.


  
    
  
  [0.5.0] - 2020-05-28



  
    
  
  Added


	New function Flop.validate!/2.
	New filter operator :in.


  
    
  
  Fixed


	Filter validation was using sortable fields instead of filterable fields.


  
    
  
  [0.4.0] - 2020-05-27



  
    
  
  Added


	Added =~ filter operator.


  
    
  
  Fixed


	Query function wasn't generating valid where clauses for filters.


  
    
  
  [0.3.0] - 2020-05-22



  
    
  
  Added


	Added a default_limit option to Flop.Schema.


  
    
  
  [0.2.0] - 2020-05-20



  
    
  
  Added


	Added a max_limit option to Flop.Schema. When set, Flop validates that the
limit and page_size parameters don't exceed the configured max limit.


  
    
  
  [0.1.0] - 2019-10-19


initial release


  

    
Schema configuration
    


  
    
  
  Basics



  
    
  
  Minimal configuration


defmodule Pet do
  use Ecto.Schema

  @derive {
    Flop.Schema,
    filterable: [:name, :species],
    sortable: [:name, :age]
  }

  schema "pets" do
    field :name, :string
    field :age, :integer
    field :species, :string
  end
end

  
    
  
  Options


Limit
@derive {
  Flop.Schema,
  filterable: [:name, :species],
  sortable: [:name, :age],
  max_limit: 100,
  default_limit: 50
}
Order
@derive {
  Flop.Schema,
  filterable: [:name, :species],
  sortable: [:name, :age],
  default_order: %{
    order_by: [:name, :age],
    order_directions: [:asc, :desc]
  }
}
Pagination types
@derive {
  Flop.Schema,
  filterable: [:name, :species],
  sortable: [:name, :age],
  pagination_types: [:first, :last],
  default_pagination_type: :first
}

  
    
  
  Alias fields



  
    
  
  Schema


defmodule Owner do
  use Ecto.Schema

  @derive {
    Flop.Schema,
    filterable: [:name],
    sortable: [:name, :pet_count],
    adapter_opts: [
      alias_fields: [:pet_count]
    ]
  }

  schema "owners" do
    field :name, :string
    has_many :pets, Pet
  end
end

  
    
  
  Query


params = %{order_by: [:pet_count]}

Owner
|> join(:left, [o], p in assoc(o, :pets), as: :pets)
|> group_by([o], o.id)
|> select(
  [o, pets: p],
  {o.id, p.id |> count() |> selected_as(:pet_count)}
)
|> Flop.validate_and_run(params, for: Owner)

  
    
  
  Compound fields



  
    
  
  Schema


defmodule User do
  use Ecto.Schema

  @derive {
    Flop.Schema,
    filterable: [:full_name],
    sortable: [:full_name],
    adapter_opts: [
      compound_fields: [
        full_name: [:family_name, :given_name]
      ]
    ]
  }

  schema "users" do
    field :family_name, :string
    field :given_name, :string
  end
end

  
    
  
  Query


params = %{
  filters: [
    %{field: :full_name, op: :ilike_and, value: "pea"}
  ]
}

Flop.validate_and_run(User, params, for: Owner)

  
    
  
  Join fields



  
    
  
  Schema


Owner
defmodule Owner do
  use Ecto.Schema

  @derive {
    Flop.Schema,
    filterable: [:name, :pet_age],
    sortable: [:name],
    adapter_opts: [
      join_fields: [
        pet_age: [
          binding: :pets,
          field: :age,
          ecto_type: :integer
        ]
      ]
    ]
  }

  schema "owners" do
    field :name, :string
    has_many :pets, Pet
  end
end
Pet
defmodule Pet do
  use Ecto.Schema

  schema "pets" do
    field :age, :integer
  end
end

  
    
  
  Query


Only filtering or sorting
params = %{
  filters: [
    %{field: :pet_age, op: :==, value: 8}
  ]
}

Owner
|> join([o], p in assoc(o, :pets), as: :pets)
|> Flop.validate_and_run(params, for: Pet)
With preload
Owner
|> join([o], p in assoc(o, :pets), as: :pets)
|> preload([pets: p], pets: p)
|> Flop.validate_and_run(params, for: Pet)

  
    
  
  Join field for nested association



  
    
  
  Schema


Owner
defmodule Owner do
  use Ecto.Schema

  @derive {
    Flop.Schema,
    filterable: [:name, :toy_description],
    sortable: [:name],
    adapter_opts: [
      join_fields: [
        pet_age: [
          binding: :toys,
          field: :description,
          ecto_type: :string,
          # only needed with cursor pagination when sorting
          # by the join field, so that Flop can find the
          # cursor value
          path: [:pets, :toys]
        ]
      ]
    ]
  }

  schema "owners" do
    field :name, :string
    has_many :pets, Pet
  end
end
Pet
defmodule Pet do
  use Ecto.Schema

  schema "pets" do
    field :age, :integer
    has_many :toys, Toy
  end
end
Toy
defmodule Toy do
  use Ecto.Schema

  schema "toys" do
    field :description, :string
  end
end

  
    
  
  Query with preload


params = %{order_by: [:toy_description]}

Owner
|> join([o], p in assoc(o, :pets), as: :pets)
|> join([pets: p], t in assoc(p, :toys), as: :toys)
|> preload([pets: p, toys: t], pets: {p, toys: t})
|> Flop.validate_and_run(params, for: Owner)

  
    
  
  Join field for subquery



  
    
  
  Schema


Owner
defmodule Owner do
  use Ecto.Schema

  @derive {
    Flop.Schema,
    filterable: [:name],
    sortable: [:name, :pet_count],
    adapter_opts: [
      join_fields: [
        pet_count: [
          binding: :pet_count,
          field: :count
        ]
      ]
    ]
  }

  schema "owners" do
    field :name, :string
    has_many :pets, Pet
  end
end
Pet
defmodule Pet do
  use Ecto.Schema

  schema "pets" do
    field :age, :integer
  end
end

  
    
  
  Query


params = %{filters: [%{field: :pet_count, op: :>, value: 2}]}

pet_count_query =
  Pet
  |> where([p], parent_as(:owner).id == p.owner_id)
  |> select([p], %{count: count(p)})

q =
  Owner
  |> from(as: :owner)
  |> join(:inner_lateral, [o], p in subquery(pet_count_query),
    as: :pet_count
  )
  |> Flop.validate_and_run(params, for: Owner)

  
    
  
  Custom fields



  
    
  
  Schema


defmodule Pet do
  use Ecto.Schema

  @derive {
    Flop.Schema,
    filterable: [:name, :human_age],
    sortable: [:name],
    adapter_opts: [
      custom_fields: [
        human_age: [
          filter: {CustomFilters, :human_age, []},
          ecto_type: :integer
        ]
      ]
    ]
  }

  schema "pets" do
    field :name, :string
    field :age, :integer
  end
end

  
    
  
  Custom filter function


defmodule CustomFilters do
  import Ecto.Query

  def human_age(q, %Flop.Filter{value: value, op: op}, _) do
    case Ecto.Type.cast(:integer, value) do
      {:ok, human_years} ->
        value_in_dog_years = round(human_years / 7)

        case op do
          :== -> where(q, [p], p == ^value_in_dog_years)
          :!= -> where(q, [p], p != ^value_in_dog_years)
          :> -> where(q, [p], p > ^value_in_dog_years)
          :< -> where(q, [p], p < ^value_in_dog_years)
          :>= -> where(q, [p], p >= ^value_in_dog_years)
          :<= -> where(q, [p], p <= ^value_in_dog_years)
        end

      :error ->
        # cannot cast filter value, ignore
        q
    end
  end
end

  
    
  
  Query


params = %{
  filters: [
    %{field: :human_age, op: :==, value: 30}
  ]
}

Flop.validate_and_run(Pet, params, for: Pet)


  

    
Flop 
    



      
Flop is a helper library for filtering, ordering and pagination with Ecto.

  
    
  
  Usage


The simplest way of using this library is just to use
Flop.validate_and_run/3 and Flop.validate_and_run!/3. Both functions
take a queryable and a parameter map, validate the parameters, run the query
and return the query results and the meta information.
iex> Flop.Repo.insert_all(MyApp.Pet, [
...>   %{name: "Harry", age: 4, species: "C. lupus"},
...>   %{name: "Maggie", age: 1, species: "O. cuniculus"},
...>   %{name: "Patty", age: 2, species: "C. aegagrus"}
...> ])
iex> params = %{order_by: ["name", "age"], page: 1, page_size: 2}
iex> {:ok, {results, meta}} =
...>   Flop.validate_and_run(
...>     MyApp.Pet,
...>     params,
...>     repo: Flop.Repo
...>   )
iex> Enum.map(results, & &1.name)
["Harry", "Maggie"]
iex> meta.total_count
3
iex> meta.total_pages
2
iex> meta.has_next_page?
true
Under the hood, these functions just call Flop.validate/2 and Flop.run/3,
which in turn calls Flop.all/3 and Flop.meta/3. If you need finer control
about if and when to execute each step, you can call those functions directly.
See Flop.Meta for descriptions of the meta fields.

  
    
  
  Global configuration


You can set some global options like the default Ecto repo via the application
environment. All global options can be overridden by passing them directly to
the functions or configuring the options for a schema module via
Flop.Schema.
import Config

config :flop, repo: MyApp.Repo
See Flop.option/0 for a description of all available options.

  
    
  
  Config modules


Instead of setting global options in the application environment, you can also
create a Flop config module. This is especially useful in an umbrella
application, or if you have multiple Repos.
defmodule MyApp.Flop do
  use Flop, repo: MyApp.Repo, default_limit: 25
end
use Flop
When you use Flop, the Flop module will define wrapper functions around
all of the Flop functions that take a query, the Flop parameters, and
options as arguments. The options passed to use Flop will be used as
default options in all the wrapper functions, but you can still override
them.

The wrapped functions are:
	Flop.all/3
	Flop.count/3
	Flop.filter/3
	Flop.meta/3
	Flop.order_by/3
	Flop.paginate/3
	Flop.query/3
	Flop.run/3
	Flop.validate_and_run/3
	Flop.validate_and_run!/3

So instead of using Flop.validate_and_run/3, you would call
MyApp.Flop.validate_and_run/3.
If you have both a config module and a global application config, Flop will
fall back to the application config if an option is not set.
See Flop.option/0 for a description of all available options.

  
    
  
  Schema options


You can set some options for a schema by deriving Flop.Schema. The options
are evaluated at the validation step.
defmodule Pet do
  use Ecto.Schema

  @derive {Flop.Schema,
           filterable: [:name, :species],
           sortable: [:name, :age],
           default_limit: 20,
           max_limit: 100}

  schema "pets" do
    field :name, :string
    field :age, :integer
    field :species, :string
    field :social_security_number, :string
  end
end
You need to pass the schema to Flop.validate/2 or any function that
includes the validation step with the :for option.
iex> params = %{"order_by" => ["name", "age"], "limit" => 5}
iex> {:ok, flop} = Flop.validate(params, for: MyApp.Pet)
iex> flop.limit
5

iex> params = %{"order_by" => ["name", "age"], "limit" => 10_000}
iex> {:error, meta} = Flop.validate(params, for: MyApp.Pet)
iex> [limit: [{msg, _}]] = meta.errors
iex> msg
"must be less than or equal to %{number}"

iex> params = %{"order_by" => ["name", "age"], "limit" => 10_000}
iex> {:error, %Flop.Meta{} = meta} =
...>   Flop.validate_and_run(
...>     MyApp.Pet,
...>     params,
...>     for: MyApp.Pet
...>   )
iex> [limit: [{msg, _}]] = meta.errors
iex> msg
"must be less than or equal to %{number}"

  
    
  
  Ordering


To add an ordering clause to a query, you need to set the :order_by and
optionally the :order_directions parameter. :order_by should be the list
of fields, while :order_directions is a list of Flop.order_direction/0.
:order_by and :order_directions are zipped when generating the ORDER BY
clause. If no order directions are given, :asc is used as default.
iex> params = %{
...>   "order_by" => ["name", "age"],
...>   "order_directions" => ["asc", "desc"]
...> }
iex> {:ok, flop} = Flop.validate(params)
iex> flop.order_by
[:name, :age]
iex> flop.order_directions
[:asc, :desc]
Flop uses these two fields instead of a keyword list, so that the order
instructions can be easily passed in a query string.

  
    
  
  Pagination


For queries using OFFSET and LIMIT, you have the choice between
page-based pagination parameters:
%{page: 5, page_size: 20}
and offset-based pagination parameters:
%{offset: 100, limit: 20}
For cursor-based pagination, you can either use :first/:after or
:last/:before. You also need to pass the :order_by parameter or set a
default order for the schema via Flop.Schema.
iex> Flop.Repo.insert_all(MyApp.Pet, [
...>   %{name: "Harry", age: 4, species: "C. lupus"},
...>   %{name: "Maggie", age: 1, species: "O. cuniculus"},
...>   %{name: "Patty", age: 2, species: "C. aegagrus"}
...> ])
iex>
iex> # forward (first/after)
iex>
iex> params = %{first: 2, order_by: [:species, :name]}
iex> {:ok, {results, meta}} = Flop.validate_and_run(MyApp.Pet, params)
iex> Enum.map(results, & &1.name)
["Patty", "Harry"]
iex> meta.has_next_page?
true
iex> end_cursor = meta.end_cursor
"g3QAAAACdwRuYW1lbQAAAAVIYXJyeXcHc3BlY2llc20AAAAIQy4gbHVwdXM="
iex> params = %{first: 2, after: end_cursor, order_by: [:species, :name]}
iex> {:ok, {results, meta}} = Flop.validate_and_run(MyApp.Pet, params)
iex> Enum.map(results, & &1.name)
["Maggie"]
iex> meta.has_next_page?
false
iex>
iex> # backward (last/before)
iex>
iex> params = %{last: 2, order_by: [:species, :name]}
iex> {:ok, {results, meta}} = Flop.validate_and_run(MyApp.Pet, params)
iex> Enum.map(results, & &1.name)
["Harry", "Maggie"]
iex> meta.has_previous_page?
true
iex> start_cursor = meta.start_cursor
"g3QAAAACdwRuYW1lbQAAAAVIYXJyeXcHc3BlY2llc20AAAAIQy4gbHVwdXM="
iex> params = %{last: 2, before: start_cursor, order_by: [:species, :name]}
iex> {:ok, {results, meta}} = Flop.validate_and_run(MyApp.Pet, params)
iex> Enum.map(results, & &1.name)
["Patty"]
iex> meta.has_previous_page?
false
By default, it is assumed that the query result is a list of maps or structs.
If your query returns a different data structure, you can pass the
:cursor_value_func option to retrieve the cursor values. See
Flop.option/0 and Flop.Cursor for more information.
You can restrict which pagination types are available. See Flop.option/0
for details.

  
    
  
  Filters


Filters can be passed as a list of maps. It is recommended to define the
filterable fields for a schema using Flop.Schema.
iex> Flop.Repo.insert_all(MyApp.Pet, [
...>   %{name: "Harry", age: 4, species: "C. lupus"},
...>   %{name: "Maggie", age: 1, species: "O. cuniculus"},
...>   %{name: "Patty", age: 2, species: "C. aegagrus"}
...> ])
iex>
iex> params = %{filters: [%{field: :name, op: :=~, value: "Mag"}]}
iex> {:ok, {results, meta}} = Flop.validate_and_run(MyApp.Pet, params)
iex> meta.total_count
1
iex> [pet] = results
iex> pet.name
"Maggie"
See Flop.Filter.op/0 for a list of all available filter operators.

  
    
  
  GraphQL and Relay


The parameters used for cursor-based pagination follow the Relay
specification, so you can just pass the arguments you get from the client on
to Flop.
Flop.Relay can convert the query results returned by
Flop.validate_and_run/3 into Edges and PageInfo formats required for
Relay connections.
For example, if you have a context module like this:
defmodule MyApp.Flora
  import Ecto.query, warn: false

  alias MyApp.Flora.Plant

  def list_plants_by_continent(%Continent{} = continent, %{} = args) do
    Plant
    |> where(continent_id: ^continent.id)
    |> Flop.validate_and_run(args, for: Plant)
  end
end
Then your Absinthe resolver for the plants connection may look something
like this:
def list_plants(args, %{source: %Continent{} = continent}) do
  with {:ok, result} <-
         Flora.list_plants_by_continent(continent, args) do
    {:ok, Flop.Relay.connection_from_result(result)}
  end
end

Functions for encoding, decoding and extracting cursor values.

Flop.FieldInfo 
    



      
Defines a struct that holds the information about a schema field.
This struct is mainly for use by adapters.

Defines a filter.

Defines a struct for holding meta information of a query result.

Helpers to turn query results into Relay formats.

Flop.Schema protocol
    



      
Flop.Schema is a protocol that allows you to customize and set query options
in your Ecto schemas.
This module allows you to define which fields are filterable and sortable, set
default and maximum limits, specify default sort orders, restrict pagination
types, and more.

  
    
  
  Usage


To utilize this protocol, derive Flop.Schema in your Ecto schema and define
the filterable and sortable fields.
defmodule MyApp.Pet do
  use Ecto.Schema

  @derive {
    Flop.Schema,
    filterable: [:name, :species],
    sortable: [:name, :age]
  }

  schema "pets" do
    field :name, :string
    field :age, :integer
    field :species, :string
  end
end
See option/0 for an overview of all available options.
@derive Flop.Schema
When you derive Flop.Schema, all the functions required for the
Flop.Schema protocol will be defined based on the options you set.

After that, you can pass the module as the :for option to Flop.validate/2.
iex> Flop.validate(%Flop{order_by: [:name]}, for: MyApp.Pet)
{:ok,
 %Flop{
   filters: [],
   limit: 50,
   offset: nil,
   order_by: [:name],
   order_directions: nil,
   page: nil,
   page_size: nil
 }}

iex> {:error, %Flop.Meta{} = meta} = Flop.validate(
...>   %Flop{order_by: [:species]}, for: MyApp.Pet
...> )
iex> meta.params
%{"order_by" => [:species], "filters" => []}
iex> meta.errors
[
  order_by: [
    {"has an invalid entry",
     [validation: :subset, enum: [:name, :age, :owner_name, :owner_age]]}
  ]
]

  
    
  
  Default and maximum limits


Define a default or maximum limit by setting the default_limit and
max_limit options while deriving Flop.Schema. Flop.validate/1 will apply
the default limit and validate the maximum limit.
@derive {
  Flop.Schema,
  filterable: [:name, :species],
  sortable: [:name, :age],
  max_limit: 100,
  default_limit: 50
}

  
    
  
  Default sort order


Specify a default sort order by setting the default_order_by and
default_order_directions options when deriving Flop.Schema. The default
values will be applied by Flop.validate/1. If no order directions are set,
:asc is the default for all fields.
@derive {
  Flop.Schema,
  filterable: [:name, :species],
  sortable: [:name, :age],
  default_order: %{
    order_by: [:name, :age],
    order_directions: [:asc, :desc]
  }
}

  
    
  
  Restricting pagination types


By default, all supported pagination types (Flop.pagination_type/0) are
enabled. If you wish to restrict the pagination type for a schema, you can
set the :pagination_types option.
@derive {
  Flop.Schema,
  filterable: [:name, :species],
  sortable: [:name, :age],
  pagination_types: [:first, :last]
}
Setting the value to nil (default) allows all pagination types.
See also Flop.option/0. 

  
    
  
  Alias fields


To sort by calculated values, you can use Ecto.Query.API.selected_as/2 in
your query, define an alias field in your schema, and add the alias field to
the list of sortable fields.
Schema:
@derive {
  Flop.Schema,
  filterable: [],
  sortable: [:pet_count],
  adapter_opts: [
    alias_fields: [:pet_count]
  ]
}
Query:
Owner
|> join(:left, [o], p in assoc(o, :pets), as: :pets)
|> group_by([o], o.id)
|> select(
  [o, pets: p],
  {o.id, p.id |> count() |> selected_as(:pet_count)}
)
|> Flop.validate_and_run(params, for: Owner)
Note that it is not possible to use field aliases in WHERE clauses, which
means you cannot add alias fields to the list of filterable fields, and you
cannot sort by an alias field if you are using cursor-based pagination.

  
    
  
  Compound fields


Sometimes you might need to apply a search term to multiple fields at once,
e.g. you might want to search in both the family name and given name field.
You can do that with Flop by defining a compound field.
@derive {
  Flop.Schema,
  filterable: [:full_name],
  sortable: [:full_name],
  adapter_opts: [
    compound_fields: [full_name: [:family_name, :given_name]]
  ]
}
This allows you to use the field name :full_name as any other field in the
filter and order parameters.

  
    
  
  Filtering


params = %{
  filters: [%{
    field: :full_name,
    op: :like,
    value: "margo"
  }]
}
This would translate to:
WHERE family_name like '%margo%' OR given_name like '%margo%'
Partial matches of the search term can be achieved with one of
the like operators.
params = %{
  filters: [%{
    field: :full_name,
    op: :ilike_and,
    value: ["margo", "martindale"]
  }]
}
or
params = %{
  filters: [%{
    field: :full_name,
    op: :ilike_and,
    value: "margo martindale"
  }]
}
This would translate to:
WHERE (family_name ilike '%margo%' OR given_name ilike '%margo%')
AND (family_name ilike '%martindale%' OR given_name ilike '%martindale%')

  
    
  
  Filter operator rules


	:=~ :like :not_like :like_and :like_or :ilike :not_ilike :ilike_and :ilike_or
If a string value is passed it will be split at whitespace
characters and each segment will be checked separately. If a list of strings is
passed the individual strings are not split. The filter matches for a value
if it matches for any of the fields.
	:empty
Matches if all fields of the compound field are nil.
	:not_empty
Matches if any field of the compound field is not nil.
	:== :!= :<= :< :>= :> :in :not_in :contains :not_contains
 These filter operators are ignored for compound fields at the moment.
This will be added in a future version.
The filter value is normalized by splitting the string at whitespaces and
joining it with a space. The values of all fields of the compound field are
split by whitespace character and joined with a space, and the resulting
values are joined with a space again.


  
    
  
  Sorting


params = %{
  order_by: [:full_name],
  order_directions: [:desc]
}
This would translate to:
ORDER BY family_name DESC, given_name DESC
Note that compound fields cannot be used as pagination cursors.

  
    
  
  Join fields


If you need to filter or order across tables, you can define join fields.
As an example, let's define these schemas:
schema "owners" do
  field :name, :string
  field :email, :string

  has_many :pets, Pet
end

schema "pets" do
  field :name, :string
  field :species, :string

  belongs_to :owner, Owner
end
And now we want to find all owners that have pets of the species
"E. africanus". To do this, first we need to define a join field on the
Owner schema.
@derive {
  Flop.Schema,
  filterable: [:pet_species],
  sortable: [:pet_species],
  adapter_opts: [
    join_fields: [
      pet_species: [
        binding: :pets,
        field: :species,
        ecto_type: :string
      ]
    ]
  ]
}
In this case, :pet_species would be the alias of the field that you can
refer to in the filter and order parameters. The options are:
	:binding - The named binding you set with the :as option in the join
statement of your query.
	:field - The field on that binding on which the filter should be applied.
	:ecto_type - The Ecto type of the field. This allows Flop to validate
filter values, and also to treat empty arrays and empty maps as empty values
depending on the type. See also Ecto type option section below.

In order to retrieve the pagination cursor value for a join field, Flop needs
to know how to get the field value from the struct that is returned from the
database. Flop.Schema.get_field/2 is used for that. By default, Flop assumes
that the binding name matches the name of the field for the association in
your Ecto schema (the one you set with has_one, has_many or belongs_to).
In the example above, Flop would try to access the field in the struct under
the path [:pets, :species].
If you have joins across multiple tables, or if you can't give the binding
the same name as the association field, you can specify the path explicitly.
@derive {
  Flop.Schema,
  filterable: [:pet_species],
  sortable: [:pet_species],
  adapter_opts: [
    join_fields: [
      pet_species: [
        binding: :pets,
        field: :species,
        path: [:pets, :species]
      ]
    ]
  ]
}
After setting up the join fields, you can write a query like this:
params = %{
  filters: [%{field: :pet_species, op: :==, value: "E. africanus"}]
}

Owner
|> join(:left, [o], p in assoc(o, :pets), as: :pets)
|> preload([pets: p], [pets: p])
|> Flop.validate_and_run!(params, for: Owner)
If your query returns data in a different format, you don't need to set the
:path option. Instead, you can pass a custom cursor value function in the
options. See Flop.Cursor.get_cursors/2 and Flop.option/0.
Note that Flop doesn't create the join clauses for you. The named bindings
already have to be present in the query you pass to the Flop functions. You
can use Flop.with_named_bindings/4 or Flop.named_bindings/3 to get the
build the join clauses needed for a query dynamically and avoid adding
unnecessary joins.

  
    
  
  Filtering by calculated values with subqueries


You can join on a subquery with a named binding and add a join field as
described above.
Schema:
@derive {
  Flop.Schema,
  filterable: [:pet_count],
  sortable: [:pet_count],
  adapter_opts: [
    join_fields: [
      pet_count: [
        binding: :pet_count,
        field: :count,
        ecto_type: :integer
      ]
    ]
  ]
}
Query:
params = %{filters: [%{field: :pet_count, op: :>, value: 2}]}

pet_count_query =
  Pet
  |> where([p], parent_as(:owner).id == p.owner_id)
  |> select([p], %{count: count(p)})

q =
  (o in Owner)
  |> from(as: :owner)
  |> join(:inner_lateral, [owner: o], p in subquery(pet_count_query),
    as: :pet_count
  )
  |> Flop.validate_and_run(params, for: Owner)

  
    
  
  Custom fields


Custom fields allow for precise control over filter queries, making it
possible to implement filter logic that the built-in filtering options cannot
satisfy.
For example, you might need to handle dates and times in a particular way that
takes into account different time zones, or perform database-specific queries
using fragments.
Custom field filters are referenced by a tuple
{mod :: module, function :: atom, opts :: keyword}. The referenced function
receives three arguments: the Ecto query, the Flop filter, and an options
keyword list.
If runtime options are necessary (like the timezone of the request or the user
ID of the current user), use the extra_opts option when calling Flop
functions.
Note that as of now, custom fields only support filtering, not sorting.
Schema:
@derive {
  Flop.Schema,
  filterable: [:inserted_at_date],
  adapter_opts: [
    custom_fields: [
      inserted_at_date: [
        filter: {CustomFilters, :date_filter, [source: :inserted_at]},
        ecto_type: :date,
        operators: [:<=, :>=]
      ]
    ]
  ]
}
If you pass the :ecto_type option like above, the filter valu